

SLOT GAME

成重昭博

MSX

今回のプログラムは読者からの投稿によるものです。作者の成重昭博君は大阪に住む13歳の中学生。なかなか手慣れたプログラムでしたが、手直しすれば、もっと楽しくなりそうな部分もあります。もちろん、このまでも十分楽しめるものですから、皆さんのソフトライブラリにぜひ加えてください。

●内容

皆さんは『片腕の泥棒』という言葉をご存知ですか。これは、スロットマシンのアダ名なのです。なにしろ、あまり考える必要のない、機械相手のゲームですから、ついつい、もう1回、もう1回ということになってしまい、気がつくとサイフの中はスッカラカンということが多いという話です。

このスロットマシンをコンピュータ化したのが、『スロットゲーム』です。本物のスロットマシンのように、コインを投入する必要はありませんが、あまりアツくなって、仕事や勉強を忘れないようにしてください。

●ゲーム方法

まずプログラムをRUNさせると、タイトル画面の後、倍率表が表示されます。この倍率表というのは絵柄の組み合わせによって、賭け金がそれぞれ何倍になって返ってくるかを示したものです。もちろん、計算はコンピュータが行ないますが、ここでどの組み合わせをねらうか、考えておきましょう。次に、画面の指示通り「SPACE」キーを押すと、画面にスロットマシンがあらわれ、ゲームの開始です。

最初に賭け金を入力します。賭け金の上限は\$10000で、それを越える額は入力できません。もちろん自分の持ち金を越える額の入力も受けつけません。最初の持ち金は\$5000

です。この時に数字キー以外は受けつけないようになっていますから、「カナ」キーが押されたままになっていないか注意してください。賭け金を訂正したい時は「BS」キーを押してください。賭け金が0に戻ります。賭け金が決ったら「RETURN」キーを押します。これでスロットが回転を始めるわけです。

スロットの回転を止めるのは「SPACE」キーです。1回押すごとに左側のスロットから順に止まります。止まった絵柄の組み合わせで配当金が決まります。当りの場合は配当金が持ち金に加算され、はずれの場合には賭け金が持ち金から差し引かれます。これで1回のゲームが終了して、また賭け金の入力となります。

ゲームは全部で10回行われ、10回目が終了した時点での持ち金の額で優劣を争います。途中で持ち金が0になると、そこでGAME OVERです。

このプログラムを実行している限り、ベスト5が記憶されています。10回目が終了した時点で、ベスト5に入ることができると、自分の名前(3文字以内)を入力することができます。入力方法は名前「RETURN」です。

途中でGAME OVERあるいは10回目までゲームが終わると、プレイヤーの得点が表示され、その後ベスト5が表示されます。ここで画面の指示に従って「SPACE」キーを押すと、再びゲームの始めに戻ります。

●ゲームのモディファイ

プログラムに少し手を加えると、ゲームのモディファイをすることができます。

☆1090行

KY\$(1) = "7" : KY\$(2) = "J" ……

でスロットの絵柄を定めています。自分なりにキャラクタを入れてみましょう。例えば
KY\$(1) = "♥" : KY\$(2) = "♣" ……
——という具合にです。

☆1100行

KY(1) = 13 : KY(2) = 1 …… KY(5) = 2

というのは、スロットの絵柄の色をそれぞれ決めていきます。KY(1)はKY\$(1)とKY(2)はKY\$(2)とそれぞれ対応しています。ですからこのプログラムでは、「7」というキャラクタは13の色ということになります。カラーコードについては、それぞれのマシンに付いている文法書を見てください。

☆1110行

MO = 5000

これが最初の持ち金です。同じ行の最後の
MX = 10000

というのが、賭け金の限度額です。

これは、あまり大幅に変えないほうが、ゲームとしては楽しいようです。また、賭け金や持ち金の額をあまり多くしてしまうと、配当金を加算した時に表示がおかしくなってしまうこともあります(数字のケタが多すぎるため)。

●注意

このプログラムは、実際に動作したものをそのままプリントアウトし、縮小、印刷したものです。普通に使用する限り、なんら問題はないはずですが(ただし、自分でモディファイした場合は別です)。

もしもエラーが発生した場合、タイプミスの可能性が非常に大きいからです、十分に注意してプログラムを入力してください。

```

1000 '*****
1010 '* *
1020 '* SLOT GAME (MSX) *
1030 '* *
1040 '* by A.Narishige *
1050 '* *
1060 '*****
1070 DIM KY$(5),KY(5),HI(5),HI$(5)
1080 FOR I=1 TO 5:HI(I)=1000:HI$(I)="MSX":NEXT I
1090 KY$(1)="7":KY$(2)="J":KY$(3)="=":KY$(4)="A":KY$(5)="●"
1100 KY(1)=13:KY(2)=1:KY(3)=4:KY(4)=9:KY(5)=2
1110 MO=5000:KA=0:KA$="":ST$="":GM=1:BA=0:NO=1:MX=10000
1120 PLAY"T180V15","T180V12"
1130 GOSUB 1680:GOSUB 1750
1140 '
1150 '*** メイン ルーチン ***
1160 '
1170 COLOR ,0,15:SCREEN 2:OPEN"GRP:"AS#1
1180 GOSUB 1940
1190 LINE(184,32)-(247,39),15,BF:PSET(136,32),15:COLOR 1:PRINT#1,USING"モチキン :$##
#####";MO
1200 PSET(136,48),15:COLOR 13:PRINT#1,USING"ハイスコア:$#####";HI(1)
1210 LINE(168,80)-(183,88),15,BF:PSET(168,80),15:PRINT#1,USING"## カイメ";NO
1220 PLAY"04L16CDEDC"
1230 LINE(136,112)-(256,191),15,BF:PSET(136,112),15:COLOR 1:PRINT#1,"イクラケマスか?"
1240 IF KA>MX OR KA>MO THEN KA=0:KA$="0":PLAY"02C64"
1250 LINE(160,128)-(255,135),15,BF:PSET(152,128),15:COLOR 4:PRINT#1,USING"$#####
##";KA
1260 ST$=INKEY$
1270 IF ST$="" THEN 1260
1280 IF ST$=CHR$(8) THEN 1330
1290 IF ST$=CHR$(13) THEN 1320
1300 IF ASC(ST$)<48 OR ASC(ST$)>57 THEN 1260
1310 KA$=KA$+ST$:KA=VAL(KA$):GOTO 1240
1320 IF KA<=0 OR KA>MO THEN PLAY"02C64":GOTO 1330 ELSE 1340
1330 KA$="":KA=0:GOTO 1240
1340 ON STRIG GOSUB 2150
1350 STRIG(0) ON
1360 ON GM GOTO 1370,1390,1410,1430
1370 L=INT(RND(-TIME)*5)+1
1380 LINE(44,64)-(51,71),15,BF:PSET(44,64),15:COLOR KY(L):PRINT#1,USING"!";KY$(L)
)
1390 C=INT(RND(-TIME)*5)+1
1400 LINE(68,64)-(75,71),15,BF:PSET(68,64),15:COLOR KY(C):PRINT#1,USING"!";KY$(C)
)
1410 R=INT(RND(-TIME)*5)+1
1420 LINE(92,64)-(99,71),15,BF:PSET(92,64),15:COLOR KY(R):PRINT#1,USING"!";KY$(R)
):GOTO 1340
1430 STRIG(0) OFF:INTERVAL OFF
1440 GOSUB 2210
1450 '
1460 '*** ケツカ ノ ヒョウシ ***
1470 '
1480 LINE(136,112)-(256,191),15,BF
1490 IF BA=0 THEN 1550
1500 PLAY"06L16CECECECECE"

```

```

1510 PSET(136,112),15:COLOR 13:PRINT#1,"アタリ !!!"
1520 PSET(136,128),15:COLOR 13:PRINT#1,USING"#####*##";KA,BA
1530 KA=KA*BA:PSET(160,140),15:PRINT#1,USING"#####";KA
1540 MO=MO+KA:FOR I=0 TO 1000:NEXT I:GOTO 1580
1550 PLAY"03L8GFEDC"
1560 PSET(136,112),15:COLOR 1:PRINT#1,USING"ハズレ -#####";KA
1570 MO=MO-KA:FOR A=0 TO 1000:NEXT
1580 '*** ケ"ム オ"ハ" - ***
1590 IF MO=0 OR NO=10 THEN 1610 ELSE 1600
1600 KA$="":KA=0:ST$="":GM=1:NO=NO+1:BA=0:GOTO 1190
1610 CLOSE#1:SCREEN 1:COLOR 15,4,7
1620 PLAY"04L8GFEDGFEDGFEDC", "03L2CCDC8"
1630 LOCATE 5,8:PRINT"=== GAME OVER ==="
1640 LOCATE 5,12:PRINT USING"アナノ トクテンハ #####";MO
1650 FOR I= 0 TO 2000:NEXT I
1660 IF MO>=HI(5) THEN 2350 ELSE GOSUB 2520:GOTO 2490
1670 /
1680 '*** ショキ カ"メン ***
1690 /
1700 SCREEN 1:COLOR 15,4,7:KEY OFF:FOR I= 1 TO 10:KEY I,"":NEXT I
1710 PLAY"04L8CDEFGABO5C", "03C202B303C6"
1720 LOCATE 5,12:PRINT"=== SLOT GAME ==="
1730 IF PLAY(0)=0 THEN RETURN ELSE 1730
1740 RETURN
1750 /
1760 CLS:COLOR 12,0,15
1770 LOCATE 8,3:PRINT"* ハ"イリツ ヒョウ *"
1780 LOCATE 8,5:PRINT KY$(1);" ";KY$(1);" ";KY$(1);" : 20ハ"イ"
1790 LOCATE 8,6:PRINT KY$(1);" ";KY$(1);" : 10ハ"イ"
1800 LOCATE 8,7:PRINT KY$(2);" ";KY$(2);" ";KY$(2);" : 15ハ"イ"
1810 LOCATE 8,8:PRINT KY$(2);" ";KY$(2);" : 8ハ"イ"
1820 LOCATE 8,9:PRINT KY$(3);" ";KY$(3);" ";KY$(3);" : 15ハ"イ"
1830 LOCATE 8,10:PRINT KY$(4);" ";KY$(4);" ";KY$(4);" : 10ハ"イ"
1840 LOCATE 8,11:PRINT KY$(4);" ";KY$(4);" : 5ハ"イ"
1850 LOCATE 8,12:PRINT KY$(5);" ";KY$(5);" ";KY$(5);" : 10ハ"イ"
1860 LOCATE 8,13:PRINT KY$(5);" ";KY$(5);" ";KY$(1);" : 8ハ"イ"
1870 LOCATE 8,14:PRINT KY$(5);" ";KY$(5);" ";KY$(2);" : 6ハ"イ"
1880 LOCATE 8,15:PRINT KY$(5);" ";KY$(5);" ";KY$(3);" : 5ハ"イ"
1890 LOCATE 8,16:PRINT KY$(5);" ";KY$(5);" ";KY$(4);" : 2ハ"イ"
1900 LOCATE 8,18:PRINT"ソノホカ : 0ハ"イ"
1910 LOCATE 6,22:PRINT"ス"ハ"°-スキ"ラ オシテク"サ"イ"
1920 IF STRIG(0)=-1 THEN RETURN ELSE 1920
1930 /
1940 '*** スロット ノ セツテイ ***
1950 /
1960 PSET(56,0),15:COLOR 4:PRINT#1,"=== SLOT GAME ==="
1970 LINE(24,24)-(119,159),6,BF
1980 LINE(32,120)-(111,151),1,BF
1990 LINE(32,32)-(111,45),13,BF
2000 LINE(32,32)-(111,45),15,B
2010 LINE(38,58)-(105,77),14,BF
2020 LINE(40,60)-(55,75),15,BF
2030 LINE(64,60)-(79,75),15,BF
2040 LINE(88,60)-(103,75),15,BF
2050 LINE(40,88)-(103,111),15,BF
2060 PSET(44,90),15:COLOR KY(1):PRINT#1,KY$(1);KY$(1);KY$(1)

```

```

2070 PSET(44,102),15:COLOR KY(2):PRINT#1,KY$(2);KY$(2);KY$(2)
2080 PSET(76,88),15:COLOR KY(3):PRINT#1,KY$(3);KY$(3);KY$(3)
2090 PSET(76,96),15:COLOR KY(4):PRINT#1,KY$(4);KY$(4);KY$(4)
2100 PSET(76,104),15:COLOR KY(5):PRINT#1,KY$(5);KY$(5);KY$(5)
2110 LINE(124,24)-(128,95),14,BF
2120 CIRCLE(126,20),4,1,0,6.28:PAINT(126,20),1
2130 RETURN
2140 '
2150 '*** スロット ストップ ***
2160 '
2170 PLAY"05L64CDE"
2180 GM=GM+1
2190 RETURN
2200 '
2210 '*** ハンテイ ***
2220 '
2230 IF L=C THEN 2240 ELSE RETURN
2240 IF C=1 AND R=1 THEN BA=20 ELSE IF C=1 THEN BA=10
2250 IF C=2 AND R=2 THEN BA=15 ELSE IF C=2 THEN BA=8
2260 IF C=3 AND R=3 THEN BA=15
2270 IF C=4 AND R=4 THEN BA=10 ELSE IF C=4 THEN BA=5
2280 IF C=5 AND R=5 THEN BA=10
2290 IF C=5 AND R=1 THEN BA=8
2300 IF C=5 AND R=2 THEN BA=6
2310 IF C=5 AND R=3 THEN BA=5
2320 IF C=5 AND R=4 THEN BA=2
2330 RETURN
2340 '
2350 '*** ハイスコア ***
2360 '
2370 B=1
2380 IF MO>=HI(B) THEN 2400
2390 B=B+1:GOTO 2380
2400 FOR I=5 TO B STEP -1
2410 HI(I)=HI(I-1)
2420 HI$(I)=HI$(I-1)
2430 NEXT I
2440 HI(B)=MO:HI$(B)="
2450 GOSUB 2520
2460 LOCATE 5,19:PRINT"アタハ ";B;"イニ ハイリマシタ!"
2470 LOCATE 0,21:INPUT"ナマイライレテクダ"サイ(3モシ)";HI$(B)
2480 GOSUB 2520
2490 LOCATE 6,22:PRINT"スハ°-スキーヲ オシテクダ"サイ"
2500 IF STRIG(0)=-1 THEN 1110 ELSE 2500
2510 '
2520 '*** ハイスコア ノ ヒョウシ ***
2530 '
2540 CLS
2550 LOCATE 7,2:PRINT"=== BEST 5 ==="
2560 LOCATE 4,6:PRINT"NO          トクテン          ナマイ"
2570 LOCATE 5,8:PRINT USING"1          $#####          & &";HI(1),HI$(1)
2580 LOCATE 5,10:PRINT USING"2          $#####          & &";HI(2),HI$(2)
2590 LOCATE 5,12:PRINT USING"3          $#####          & &";HI(3),HI$(3)
2600 LOCATE 5,14:PRINT USING"4          $#####          & &";HI(4),HI$(4)
2610 LOCATE 5,16:PRINT USING"5          $#####          & &";HI(5),HI$(5)
2620 RETURN

```